

## excel yourself

Autoren: Ralf Sowa, Christian Hapke

Beachten Sie unsere [Hinweise](#) und [Nutzungsbedingungen](#). Vorgestellte Musterlösungen basieren auf MS-Excel® 2003; sie gelten ausschließlich für aufgezeigte Beispieldaten. Bitte melden Sie uns etwaige Fehler in unseren Informationen – Ihr Feedback ist willkommen: [urs.toolbox@urs-beratung.de](mailto:urs.toolbox@urs-beratung.de)

Eine Übersicht zu unseren Excel-Informationen finden Sie hier: [www.urs-beratung.de/toolbox.htm](http://www.urs-beratung.de/toolbox.htm)

---

### Excel - Funktionen 6

INDEX - VERGLEICH - VERWEIS

---

## INDEX

Die Funktion INDEX steht in zwei *Varianten* zur Verfügung:

- INDEX ( Matrix; Zeile; Spalte )
- INDEX ( Bezug; Zeile; Spalte; Bereich )

### INDEX ( Matrix; Zeile; Spalte )

Mit der Funktion INDEX wird der Inhalt einer Zelle ausgelesen.

Die Funktion *dieses Typs* erwartet drei Argumente:

- (1) den Bereich, aus dem die Matrix *für diesen Zugriff* besteht.

Der Bereich muss nicht dem vollständigen Datenbereich entsprechen, kann also ein Auszug daraus sein! Siehe folgende Beispiele.

- (2) die Angabe der Zeile (x-te Zeile bezogen auf die definierte Matrix)
- (3) die Angabe der Spalte (x-te Spalte bezogen auf die definierte Matrix)

	A	B	C	D	E	F	G
1	WP_Bezeichnung	16.10.07	17.10.07	18.10.07			
2	Adidas	44,12	44,40	44,51		75,00	=INDEX(A:D;9;3)
3	Allianz	161,12	162,98	162,49			
4	BASF	95,38	96,94	96,50		75,00	=INDEX(A7:C11;3;3)
5	Bayer	58,67	58,85	60,25			
6	BMW	46,79	46,67	46,78		75,00	=INDEX(B6:D10;4;2)
7	Commerzbank	31,61	31,57	31,58			
8	Continental	98,58	99,39	100,70			
9	Daimler	75,63	75,00	74,55			
10	Deutsche Bank	91,14	91,83	91,89			
11	Deutsche Börse	104,37	105,14	105,66			
12	Deutsche Post	21,64	21,69	21,69			
13	Deutsche Telekom	13,49	13,44	13,38			
14	Dt. Postbank	53,75	53,17	53,43			

F2: Hier wurde die Matrix mit A:D bestimmt. Bestimmt wurden die neunte Zeile des Bereichs (hier übereinstimmend mit der Zeilennummer) sowie die dritte Spalte (hier Spalte C). Ergebnis ist der Inhalt von Zelle C9.

F4: Als Matrix wurde hier der Bereich A7:C11 gewählt. Bestimmt wurde die dritte Zeile: die erste ist die A7, die zweite A8, die dritte A9 – jetzt noch die dritte Spalte (die erste ist A9, die zweite B9 und die dritte führt zu Zelle C9).

F6: Als Matrix wurde hier B6:D10 gewählt. Bestimmt ist dessen vierte Zeile (6-7-8 → 9) und zweite Spalte (B → C). Ergebnis ist ebenfalls der Inhalt aus Zelle C9.

Besteht die Matrix aus nur einer Zeile (z. B. A1:D1) oder aus nur einer Spalte (z. B. A2:A14), ist die Angabe von Zeilen- und Spaltenposition nicht erforderlich (jedoch nicht schädlich):

	A	B	C	D	E	F	G
1	WP_Bezeichnung	16.10.07	17.10.07	18.10.07			
2	Adidas	44,12	44,40	44,51		17.10.2007	=INDEX(A1:D1;3)
3	Allianz	161,12	162,98	162,49		17.10.2007	=INDEX(A1:D1;1;3)
4	BASF	95,38	96,94	96,50			
5	Bayer	58,67	58,85	60,25		Daimler	=INDEX(A2:A14;8)
6	BMW	46,79	46,67	46,78		Daimler	=INDEX(A2:A14;8;1)
7	Commerzbank	31,61	31,57	31,58			
8	Continental	98,58	99,39	100,70			
9	Daimler	75,63	75,00	74,55			
10	Deutsche Bank	91,14	91,83	91,89			
11	Deutsche Börse	104,37	105,14	105,66			
12	Deutsche Post	21,64	21,69	21,69			
13	Deutsche Telekom	13,49	13,44	13,38			
14	Dt. Postbank	53,75	53,17	53,43			

**INDEX ( Bezug; Zeile; Spalte; Bereich )**

Auch in ihrer zweiten *Variante* gibt die Funktion INDEX den Inhalt einer Zelle aus.

Die Funktion *dieses Typs* erwartet vier Argumente (eins mehr):

- (1) der *Bezug* besteht aus einem oder mehreren Zellbereichen

In der Version INDEX(**Matrix**;Zeile;Spalte) ermöglicht die Funktion die Angabe *nur einer* Matrix – die Formel sieht beispielsweise so aus (die Formel liefert den Inhalt aus Zelle C7):

**=INDEX ( A5:D15 ; 3 ; 3 )**

In der Version INDEX(**Bezug**;Zeile;Spalte;**Bereich**) können nun *mehrere* Matrizen definiert werden – die Formel sieht beispielsweise so aus (auch diese drei Formeln liefern jeweils den Inhalt aus Zelle C7):

**=INDEX ( ( A5:D15 ; B7:C8 ; C:F ) ; 3 ; 3 ; 1 )**

**=INDEX ( ( A5:D15 ; B7:C8 ; C:F ) ; 1 ; 2 ; 2 )**

**=INDEX ( ( A5:D15 ; B7:C8 ; C:F ) ; 7 ; 1 ; 3 )**

- (2) die Angabe der Zeile (x-te Zeile bezogen auf die definierte Matrix)

- (3) die Angabe der Spalte (x-te Spalte bezogen auf die definierte Matrix)

- (4) der *Bereich* bestimmt die Angabe des *Bezugs* (1). Steht hier eine 1, gilt der zuerst definierte Bezug, steht hier eine 2 der als zweites definierte Bezug usw.

In folgender Formel wird durch Wahl des *Bereichs* 1 bestimmt, dass der erste *Bezug* gilt:

**=INDEX ( ( A5:D15 ; B7:C8 ; C:F ) ; 3 ; 3 ; 1 )**

In dieser Formel wird nur der *Bereich* 3 bestimmt – der dritte *Bezug*:

**=INDEX ( ( A5:D15 ; B7:C8 ; C:F ) ; 1 ; 7 ; 3 )**

	A	B	C	D	E	F	G
1	WP_Bezeichnung	16.10.07	17.10.07	18.10.07			
2	Adidas	44,12	44,40	44,51		75,00	=INDEX(A:D;9;3)
3	Allianz	161,12	162,98	162,49		75,00	=INDEX((A:D;A7:C11;B6:D10);9;3;1)
4	BASF	95,38	96,94	96,50			
5	Bayer	58,67	58,85	60,25		75,00	=INDEX(A7:C11;3;3)
6	BMW	46,79	46,67	46,78		75,00	=INDEX((A:D;A7:C11;B6:D10);3;3;2)
7	Commerzbank	31,61	31,57	31,58			
8	Continental	98,58	99,39	100,70		75,00	=INDEX(B6:D10;4;2)
9	Daimler	75,63	75,00	74,55		75,00	=INDEX((A:D;A7:C11;B6:D10);4;2;3)
10	Deutsche Bank	91,14	91,83	91,89			
11	Deutsche Börse	104,37	105,14	105,66			
12	Deutsche Post	21,64	21,69	21,69			
13	Deutsche Telekom	13,49	13,44	13,38			
14	Dt. Postbank	53,75	53,17	53,43			

G2 / G5 / G8: Dies sind die bereits bekannten Formeln von oben – die INDEX-Funktion in ihrer ersten Variante INDEX(Matrix;Zeile;Spalte).

G3 / G6 / G9: Hier nun die Formeln für die INDEX-Funktion in ihrer zweiten Variante INDEX(Bezug;Zeile;Spalte;Bereich). – Beachten Sie, dass für alle drei Formeln ein identischer Bezug bestimmt wurde – darin besteht der Zweck der Funktion in dieser Variante, wenngleich wir hier nicht näher darauf eingehen.

## VERGLEICH

Die Funktion VERGLEICH ermittelt die Zeilen- oder Spaltenzahl (x-te Zeile oder x-te Spalte). Sie verwendet drei Argumente:

- (1) das Suchkriterium
- (2) den Bereich, in dem gesucht werden soll
- (3) die Vergleichstypangabe 1, 0 oder -1 (fehlt diese Angabe, wird sie mit 1 angenommen)

	A	B	C	D	E	F	G	H	I
1	WP_ Bezeichnung	12.10.07	15.10.07	16.10.07	17.10.07	18.10.07			
2	Adidas	44,44	44,20	44,12	44,40	44,51		17.10.07	
3	Allianz	162,50	162,50	161,12	162,98	162,49		5 =VERGLEICH(H2;1:1;0)	
4	BASF	95,61	95,85	95,38	96,94	96,50			
5	Bayer	57,68	58,98	58,67	58,85	60,25		Daimler	
6	BMW	47,49	47,30	46,79	46,67	46,78		9 =VERGLEICH(H5;A:A;0)	
7	Commerzbank	31,69	32,04	31,61	31,57	31,58			
8	Continental	99,37	99,33	98,58	99,39	100,70			
9	Daimler	75,05	75,35	75,63	75,00	74,55		75 =INDEX(A:F;VERGLEICH(H5;A:A;0);VERGLEICH(H2;1:1;0))	
10	Deutsche Bank	93,25	93,89	91,14	91,83	91,89		75 =SVERWEIS(H5;A:F;VERGLEICH(H2;1:1;0);0)	
11	Deutsche Börse	106,96	107,69	104,37	105,14	105,66		75 =WVERWEIS(H2;A:F;VERGLEICH(H5;A:A;0);0)	
12	Deutsche Post	22,13	22,14	21,64	21,69	21,69			
13	Deutsche Telekom	13,86	13,73	13,49	13,44	13,38			
14	Dt. Postbank	54,50	54,48	53,75	53,17	53,43			

H3: In der gesamten Zeile 1 wird die x-Spalte ermittelt, in der sich der Wert „17.10.2007“ befindet. Hier ist es die 5, was hier der Spalte E entspricht.

H5: In der gesamten Spalte A wird die x-Spalte ermittelt, in der sich der Wert „Daimler“ befindet. Hier ist es die 9, was hier der Zeile 9 entspricht.

Das hat natürlich einen Zweck: In Zellen H9:H11 wird mit der Funktionen INDEX, SVERWEIS und WVERWEIS der Schnittpunkt dieser zwei Werte in der Matrix ausgelesen.

### Vergleichstyp

Der Vergleichstyp kann die Zahlen 1 oder 0 oder -1 annehmen. Fehlt er, erfolgt die Berechnung wie für die Angabe der 1.

Vergleichstyp 0:

Hier wird der erste Wert zurückgegeben, der dem Suchkriterium entspricht – bei mehreren Werten, die dem Suchkriterium entsprechen, wird der erste Wert zurückgegeben. Die Daten brauchen nicht sortiert sein. Wird keine Übereinstimmung gefunden, erfolgt eine Fehlermeldung.

Vergleichstyp 1:

Es wird der übereinstimmende oder *nächst kleinere* Wert geliefert. Dafür müssen die Daten nach dem Suchkriterium *aufsteigend* sortiert sein. Enthält eine Tabelle die Werte 1 und 2 und 3, wird für das Suchkriterium 2,5 der Wert 2 geliefert.

Vergleichstyp -1:

Es wird der übereinstimmende oder *nächst größere* Wert geliefert. Dafür müssen die Daten nach dem Suchkriterium *absteigend* sortiert sein. Enthält eine Tabelle die Werte 3 und 2 und 1, wird für das Suchkriterium 2,5 der Wert 3 geliefert.

Die folgende Tabelle zeigt die Wirkungen der Vergleichstypen auf:

	A	B	C	D	E
1	Pos				
2	1	10		20	
3	2	20		2 =VERGLEICH(D2;B2:B5;0)	
4	3	30		2 =VERGLEICH(D2;B2:B5;1)	
5	4	40		#NV =VERGLEICH(D2;B2:B5;-1)	
6					
7	1	40		20	
8	2	30		3 =VERGLEICH(D7;B7:B10;0)	
9	3	20		#NV =VERGLEICH(D7;B7:B10;1)	
10	4	10		3 =VERGLEICH(D7;B7:B10;-1)	
11					
12	1	10		25	
13	2	20		#NV =VERGLEICH(D12;B12:B15;0)	
14	3	30		2 =VERGLEICH(D12;B12:B15;1)	
15	4	40		#NV =VERGLEICH(D12;B12:B15;-1)	
16					
17	1	40		25	
18	2	30		#NV =VERGLEICH(D17;B17:B20;0)	
19	3	20		#NV =VERGLEICH(D17;B17:B20;1)	
20	4	10		2 =VERGLEICH(D17;B17:B20;-1)	

## VERWEIS

Die Funktion VERWEIS steht in zwei *Varianten* zur Wahl:

- VERWEIS(Suchkriterium; Suchvektor; Ergebnisvektor)
- VERWEIS(Suchkriterium; Matrix)

### VERWEIS(Suchkriterium; Suchvektor; Ergebnisvektor)

*Vektor* ist eine Spalte bzw. eine Zeile mit identischer Zellenanzahl. Das ist wichtig: Es kann nur eine Spalte (über mehrere Zeilen) bzw. nur eine Zeile (über mehrere Spalten) ausgewählt werden!

In folgenden Musterformeln ist F5 jeweils das *Suchkriterium*, die blau markierten Zellen bilden den *Suchvektor*, die rot markierten den *Ergebnisvektor*.

**= VERWEIS ( F5 ; D:D ; B:B )**

*Eine* Spalte für den *Suchvektor* (D:D) und *eine* Spalte für den *Ergebnisvektor* (B:B), jeweils die gesamte Spalte umfassend.

**= VERWEIS ( F5 ; D2:D12 ; B2:B12 )**

*Eine* Spalte für den *Suchvektor* (D2:D12) und *eine* Spalte für den *Ergebnisvektor* (B2:B12), hier jeweils auf die Zellen der Zeilen 2 bis 12 begrenzt.

**= VERWEIS ( F5 ; A2:D2 ; A5:D5 )**

*Eine* Zeile für den *Suchvektor* (A2:D2) und *eine* Zeile für den *Ergebnisvektor* (A5:D5), hier jeweils auf die Zellen der Spalten A bis D begrenzt.

Im Gegensatz zu SVERWEIS ist die Funktion VERWEIS *dieses* Typs in der Lage, **Werte links des Suchkriteriums** aus einem Datenbereich zurückzugeben.

Weil VERWEIS aber über keinen definierbaren *Vergleichstypen* verfügt, gilt die *einheitliche* Regel: Die Datenbank muss **nach dem Suchkriterium aufsteigend sortiert** sein! (Ist das nicht der Fall, dann greifen Sie zur Auswertung mit den Funktionen INDEX mit VERGLEICH.)

	A	B	C	D	E	F	G
1	16.10.07	17.10.07	18.10.07	WP_Bezeichnung			
2	44,12	44,40	44,51	Adidas		17.10.07	
3	161,12	162,98	162,49	Allianz		=VERGLEICH(F2;1;1;0)	
4	95,38	96,94	96,50	BASF			
5	58,67	58,85	60,25	Bayer		Daimler	
6	46,79	46,67	46,78	BMW		75,00	=VERWEIS(F5;D:D;B:B)
7	31,61	31,57	31,58	Commerzbank		75,00	=VERWEIS(F5;D2:D14;B2:B14)
8	98,58	99,39	100,70	Continental			
9	75,63	75,00	74,55	Daimler			
10	91,14	91,83	91,89	Deutsche Bank		75,00	=VERWEIS(F5;D:D;WAHL(VERGLEICH(F2;1;1;0);A:A;B:B;C:C))
11	104,37	105,14	105,66	Deutsche Börse			
12	21,64	21,69	21,69	Deutsche Post			
13	13,49	13,44	13,38	Deutsche Telekom			
14	53,75	53,17	53,43	Dt. Postbank			

F6: Das Suchkriterium aus F5 (Daimler) wird in Spalte D gesucht, der zurück zu gebende Wert stammt aus Spalte B.

F7: Die angegebene Formel ist nicht empfehlenswert. Sie soll hier lediglich eine Möglichkeit aufzeigen, VERWEIS und VERGLEICH zu verbinden. Eine Kombination von INDEX mit VERGLEICH ist hier eindeutig vorzuziehen!

### VERWEIS(Suchkriterium; Matrix)

In dieser Variante entspricht die Funktion VERWEIS etwa dem SVERWEIS und WVERWEIS, wobei hier keine Angabe der Spalten- oder Zeilenposition erfolgt – VERWEIS gibt stets den Wert aus der letzten Spalte bzw. der letzten Zeile zurück.

	A	B	C	D	E	F	G	H	I
1	WP_Bezeichnung	12.10.07	15.10.07	16.10.07	17.10.07	18.10.07			
2	Adidas	44,44	44,20	44,12	44,40	44,51			
3	Allianz	162,50	162,50	161,12	162,98	162,49			
4	BASF	95,61	95,85	95,38	96,94	96,50			
5	Bayer	57,68	58,98	58,67	58,85	60,25		Daimler	
6	BMW	47,49	47,30	46,79	46,67	46,78		75,00	=VERWEIS(H5;A:E)
7	Commerzbank	31,69	32,04	31,61	31,57	31,58			
8	Continental	99,37	99,33	98,58	99,39	100,70			
9	Daimler	75,05	75,35	75,63	75,00	74,55			
10	Deutsche Bank	93,25	93,89	91,14	91,83	91,89			
11	Deutsche Börse	106,96	107,69	104,37	105,14	105,66			
12	Deutsche Post	22,13	22,14	21,64	21,69	21,69			
13	Deutsche Telekom	13,86	13,73	13,49	13,44	13,38			
14	Dt. Postbank	54,50	54,48	53,75	53,17	53,43			

H6: Als Suchkriterium ist der Inhalt von Zelle H5 („Daimler“) definiert. In der Funktion VERWEIS ist der Bereich mit A:E gewählt. Das Suchkriterium wird hier in der ersten angegebenen Spalte (A) gesucht (entspricht dem Suchvektor), das Ergebnis wird aus der letzten Spalte des Bereichs (E) geliefert (entspricht dem Ergebnisvektor).